NAG Library Routine Document

F11MLF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F11MLF computes the 1-norm, the ∞-norm or the maximum absolute value of the elements of a real, square, sparse matrix which is held in compressed column (Harwell–Boeing) format.

2 Specification

```
SUBROUTINE F11MLF(NORM, ANORM, N, ICOLZP, IROWIX, A, IFAIL)
INTEGER           N, ICOLZP(*), IROWIX(*), IFAIL
double precision  ANORM, A(*)
CHARACTER*1       NORM
```

3 Description

F11MLF computes various quantities relating to norms of a real, sparse n by n matrix A presented in compressed column (Harwell–Boeing) format.

4 References

None.

5 Parameters

1: NORM – CHARACTER*1 *Input*

On entry: specifies the value to be returned in ANORM.

NORM = '1' or 'O'

The 1-norm $\|A\|_1$ of the matrix is computed, that is $\max\limits_{1\le j\le n}\sum\limits_{i=1}^{n}\left|A_{ij}\right|$.

NORM = 'I'

The ∞-norm $\|A\|_\infty$ of the matrix is computed, that is $\max\limits_{1\le i\le n}\sum\limits_{j=1}^{n}\left|A_{ij}\right|$.

NORM = 'M'

The value $\max\limits_{1\le i,j\le n}\left|A_{ij}\right|$ (not a norm).

Constraint: NORM = '1', 'O', 'I' or 'M'.

2: ANORM – ***double precision*** *Output*

On exit: the computed quantity relating the matrix.

3: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N $\geq$ 0.

4: ICOLZP($*$) – INTEGER array *Input*

Note: the dimension of the array ICOLZP must be at least N + 1.

On entry: ICOLZP(i) contains the index in A of the start of a new column. See Section 2.1.3 in the F11 Chapter Introduction.

5: IROWIX($*$) – INTEGER array *Input*

Note: the dimension of the array IROWIX must be at least ICOLZP(N + 1) − 1, the number of nonzeros of the sparse matrix A.

On entry: the row index array of the sparse matrix A.

6: A($*$) – ***double precision*** array *Input*

Note: the dimension of the array A must be at least ICOLZP(N + 1) − 1, the number of nonzeros of the sparse matrix A.

On entry: the array of nonzero values in the sparse matrix A.

7: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value −1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value −1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, NORM ≠ '1', 'O', 'I' or 'M',
or N < 0.

IFAIL = 301

Unable to allocate required internal workspace.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example computes norms and maximum absolute value of the matrix A, where

$$A = \begin{pmatrix} 2.00 & 1.00 & 0 & 0 & 0 \\ 0 & 0 & 1.00 & -1.00 & 0 \\ 4.00 & 0 & 1.00 & 0 & 1.00 \\ 0 & 0 & 0 & 1.00 & 2.00 \\ 0 & -2.00 & 0 & 0 & 3.00 \end{pmatrix}.$$

9.1 Program Text

```
*     F11MLF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          LA, NMAX
      PARAMETER        (LA=10000,NMAX=1000)
*     .. Local Scalars ..
      DOUBLE PRECISION ANORM
      INTEGER          I, IFAIL, N, NNZ
      CHARACTER        NORM
*     .. Local Arrays ..
      DOUBLE PRECISION A(LA)
      INTEGER          ICOLZP(NMAX+1), IROWIX(LA)
*     .. External Subroutines ..
      EXTERNAL         F11MLF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F11MLF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
*
*     Read order of matrix and number of right hand sides
*
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read the matrix A
*
         DO 20 I = 1, N + 1
            READ (NIN,*) ICOLZP(I)
   20    CONTINUE
         NNZ = ICOLZP(N+1) - 1
         DO 40 I = 1, NNZ
            READ (NIN,*) A(I), IROWIX(I)
   40    CONTINUE
*
*        Calculate 1-norm
*
         NORM = '1'
         IFAIL = 1
         CALL F11MLF(NORM,ANORM,N,ICOLZP,IROWIX,A,IFAIL)
         IF (IFAIL.EQ.0) THEN
*
*           Output norm
*
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'One-norm'
            WRITE (NOUT,'(F7.3)') ANORM
*
*           Calculate M-norm
*
            NORM = 'M'
            IFAIL = 1
            CALL F11MLF(NORM,ANORM,N,ICOLZP,IROWIX,A,IFAIL)
            IF (IFAIL.EQ.0) THEN
*
*              Output norm
*
```

```
               WRITE (NOUT,*)
               WRITE (NOUT,*) 'Max'
               WRITE (NOUT,'(F7.3)') ANORM
*
*              Calculate I-norm
*
               NORM = 'I'
               IFAIL = 1
               CALL F11MLF(NORM,ANORM,N,ICOLZP,IROWIX,A,IFAIL)
               IF (IFAIL.EQ.0) THEN
*
*                 Output norm
*
                  WRITE (NOUT,*)
                  WRITE (NOUT,*) 'Infinity-norm'
                  WRITE (NOUT,'(F7.3)') ANORM
*
               END IF
            END IF
         END IF
         IF (IFAIL.NE.0) THEN
            WRITE (NOUT,99999) IFAIL
         END IF
      END IF
*
99999 FORMAT (1X,/1X,' ** F11MLF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

```
F11MLF Example Program Data
  5  N
 1
 3
 5
 7
 9
 12   ICOLZP(I) I=1,..,N+1
  2.   1
  4.   3
  1.   1
 -2.   5
  1.   2
  1.   3
 -1.   2
  1.   4
  1.   3
  2.   4
  3.   5    A(I), IROWIX(I) I=1,NNZ
```

9.3 Program Results

```
 F11MLF Example Program Results

 One-norm
  6.000

 Max
  4.000

 Infinity-norm
  6.000
```